```
Ω = {{0, 0, 1, -i}, {0, 0, i, -1}, {1, -i, 0, 0}, {i, -1, 0, 0}};
MatrixForm[Ω](* Observable *)
```

$$\begin{pmatrix} 0 & 0 & 1 & -i \\ 0 & 0 & i & -1 \\ 1 & -i & 0 & 0 \\ i & -1 & 0 & 0 \end{pmatrix}$$

```
ES = Eigensystem[Ω]
```

$$\left\{\left\{-\sqrt{2}, -\sqrt{2}, \sqrt{2}, \sqrt{2}\right\}, \left\{\left\{\frac{i}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{-\frac{1}{\sqrt{2}}, -\frac{i}{\sqrt{2}}, 1, 0\right\}, \left\{-\frac{i}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{i}{\sqrt{2}}, 1, 0\right\}\right\}\right\}$$

```
V = ES[[2]](* Not necesarily orthonormal basis *)
MatrixForm[V]
```

$$\left\{\left\{\frac{i}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{-\frac{1}{\sqrt{2}}, -\frac{i}{\sqrt{2}}, 1, 0\right\}, \left\{-\frac{i}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{i}{\sqrt{2}}, 1, 0\right\}\right\}$$

$$\begin{pmatrix} \frac{i}{\sqrt{2}} & \frac{1}{\sqrt{2}} & 0 & 1 \\ -\frac{1}{\sqrt{2}} & -\frac{i}{\sqrt{2}} & 1 & 0 \\ -\frac{i}{\sqrt{2}} & -\frac{1}{\sqrt{2}} & 0 & 1 \\ \frac{1}{\sqrt{2}} & \frac{i}{\sqrt{2}} & 1 & 0 \end{pmatrix}$$

```
VP = V[[{3, 4}]] (* Basis for  Sqrt(2) eigenspace *)
(* MatrixForm[VP]*)
VM = V[[{1, 2}]] (* Basis for -Sqrt(2) eigenspace *)
```

$$\left\{\left\{-\frac{i}{\sqrt{2}}, -\frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{\frac{1}{\sqrt{2}}, \frac{i}{\sqrt{2}}, 1, 0\right\}\right\}$$

$$\left\{\left\{\frac{i}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0, 1\right\}, \left\{-\frac{1}{\sqrt{2}}, -\frac{i}{\sqrt{2}}, 1, 0\right\}\right\}$$

```
VPO = Orthogonalize[VP](* Orthonormal basis for  Sqrt(2) eigenspace *)
(*MatrixForm[VP]*)
VMO = Orthogonalize[VM](* Orthonormal basis for -Sqrt(2) eigenspace *)
```

$$\left\{\left\{-\frac{i}{2}, -\frac{1}{2}, 0, \frac{1}{\sqrt{2}}\right\}, \left\{\frac{1}{2}, \frac{i}{2}, \frac{1}{\sqrt{2}}, 0\right\}\right\}$$

$$\left\{\left\{\frac{i}{2}, \frac{1}{2}, 0, \frac{1}{\sqrt{2}}\right\}, \left\{-\frac{1}{2}, -\frac{i}{2}, \frac{1}{\sqrt{2}}, 0\right\}\right\}$$

```
ketw1 = Transpose[{VPO[[1]]}]; MatrixForm[ketw1]
ketw2 = Transpose[{VPO[[2]]}]; MatrixForm[ketw2]
ketw3 = Transpose[{VMO[[1]]}]; MatrixForm[ketw3]
ketw4 = Transpose[{VMO[[2]]}]; MatrixForm[ketw4]
```

$$\begin{pmatrix} -\frac{i}{2} \\ -\frac{1}{2} \\ 0 \\ \frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} \\ \frac{i}{2} \\ \frac{1}{\sqrt{2}} \\ 0 \end{pmatrix}$$

$$\begin{pmatrix} \frac{i}{2} \\ \frac{1}{2} \\ 0 \\ \frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{1}{2} \\ -\frac{i}{2} \\ \frac{1}{\sqrt{2}} \\ 0 \end{pmatrix}$$

```
braw1 = ketw1†; MatrixForm[braw1]
braw2 = ketw2†; MatrixForm[braw2]
braw3 = ketw3†; MatrixForm[braw3]
braw4 = ketw4†; MatrixForm[braw4]
```

$$\begin{pmatrix} \frac{i}{2} & -\frac{1}{2} & 0 & \frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} & -\frac{i}{2} & \frac{1}{\sqrt{2}} & 0 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{i}{2} & \frac{1}{2} & 0 & \frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{1}{2} & \frac{i}{2} & \frac{1}{\sqrt{2}} & 0 \end{pmatrix}$$

```
P1 = ketw1.braw1; MatrixForm[P1]
P2 = ketw2.braw2; MatrixForm[P2]
P3 = ketw3.braw3; MatrixForm[P3]
P4 = ketw4.braw4; MatrixForm[P4]
```

$$\begin{pmatrix} \frac{1}{4} & \frac{i}{4} & 0 & -\frac{i}{2\sqrt{2}} \\ -\frac{i}{4} & \frac{1}{4} & 0 & -\frac{1}{2\sqrt{2}} \\ 0 & 0 & 0 & 0 \\ \frac{i}{2\sqrt{2}} & -\frac{1}{2\sqrt{2}} & 0 & \frac{1}{2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{4} & -\frac{i}{4} & \frac{1}{2\sqrt{2}} & 0 \\ \frac{i}{4} & \frac{1}{4} & \frac{i}{2\sqrt{2}} & 0 \\ \frac{1}{2\sqrt{2}} & -\frac{i}{2\sqrt{2}} & \frac{1}{2} & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{4} & \frac{i}{4} & 0 & \frac{i}{2\sqrt{2}} \\ -\frac{i}{4} & \frac{1}{4} & 0 & \frac{1}{2\sqrt{2}} \\ 0 & 0 & 0 & 0 \\ -\frac{i}{2\sqrt{2}} & \frac{1}{2\sqrt{2}} & 0 & \frac{1}{2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{4} & -\frac{i}{4} & -\frac{1}{2\sqrt{2}} & 0 \\ \frac{i}{4} & \frac{1}{4} & -\frac{i}{2\sqrt{2}} & 0 \\ -\frac{1}{2\sqrt{2}} & \frac{i}{2\sqrt{2}} & \frac{1}{2} & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix}$$

```
P1.P1 == P1(*Check to see if P1 is really a projector*)
P2.P2 == P2(*Check to see if P2 is really a projector*)
P3.P3 == P3(*Check to see if P3 is really a projector*)
P4.P4 == P4(*Check to see if P4 is really a projector*)
```

True

True

True

True

```
P1.P2 == DiagonalMatrix[{0, 0, 0, 0}] (*Verifying orthogonality*)
```

True

```
Ω == (√2) * P1 + (√2) * P2 + (-√2) * P3 + (-√2) * P4 (*Checking*)
```

True

```
P1 + P2 + P3 + P4 == DiagonalMatrix[{1, 1, 1, 1}]
```

True

```
PP = P1 + P2; MatrixForm[PP] (* Projector for Sqrt(2) eigenspace*)
PM = P3 + P4; MatrixForm[PM](* Projector for -Sqrt(2) eigenspace*)
```

$$\begin{pmatrix} \frac{1}{2} & 0 & \frac{1}{2\sqrt{2}} & -\frac{i}{2\sqrt{2}} \\ 0 & \frac{1}{2} & \frac{i}{2\sqrt{2}} & -\frac{1}{2\sqrt{2}} \\ \frac{1}{2\sqrt{2}} & -\frac{i}{2\sqrt{2}} & \frac{1}{2} & 0 \\ \frac{i}{2\sqrt{2}} & -\frac{1}{2\sqrt{2}} & 0 & \frac{1}{2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2} & 0 & -\frac{1}{2\sqrt{2}} & \frac{i}{2\sqrt{2}} \\ 0 & \frac{1}{2} & -\frac{i}{2\sqrt{2}} & \frac{1}{2\sqrt{2}} \\ -\frac{1}{2\sqrt{2}} & \frac{i}{2\sqrt{2}} & \frac{1}{2} & 0 \\ -\frac{i}{2\sqrt{2}} & \frac{1}{2\sqrt{2}} & 0 & \frac{1}{2} \end{pmatrix}$$

```
PP.PP == PP(*Checing to see that PP is a projector*)
PM.PM == PM(*Checking to see that PM is a projector*)
```

True

True

```
ketΨ = Transpose[{Normalize[({1, i, 0, -1})]}]; MatrixForm[ketΨ]
(*We will measure this state wrt Ω*)
braΨ = ketΨ†; MatrixForm[braΨ]
```

$$\begin{pmatrix} \frac{1}{\sqrt{3}} \\ \frac{i}{\sqrt{3}} \\ 0 \\ -\frac{1}{\sqrt{3}} \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{\sqrt{3}} & -\frac{i}{\sqrt{3}} & 0 & -\frac{1}{\sqrt{3}} \end{pmatrix}$$

```
p1 = Expand[ (braΨ.PP.ketΨ)[[1]][[1]] ](* Probability of reading the eigenvalue √2*)
p2 = Expand[ (braΨ.PM.ketΨ)[[1]][[1]] ](* Probability of reading the eigenvalue -√2*)
```

$\frac{1}{2}$

$\frac{1}{2}$

```
p1 + p2 == 1(*Check*)
```

True

```
ketΨP = Map[Simplify, Transpose[ { Normalize[ Transpose[PP.ketΨ][[1]] ] } ]];
MatrixForm[ketΨP](* Resulting state if egenvaue √2 is read*)
ketΨM = Map[Simplify, Transpose[ { Normalize[ Transpose[PM.ketΨ][[1]] ] } ]];
MatrixForm[ketΨM]
(* Resulting state if egenvaue -√2 is read*)
```

$$\begin{pmatrix} \frac{i+\sqrt{2}}{2\sqrt{3}} \\ \frac{2i+\sqrt{2}}{2\sqrt{6}} \\ \frac{1}{\sqrt{3}} \\ -\frac{1}{\sqrt{6}} \end{pmatrix}$$

$$\begin{pmatrix} \frac{-i+\sqrt{2}}{2\sqrt{3}} \\ -\frac{-2i+\sqrt{2}}{2\sqrt{6}} \\ -\frac{1}{\sqrt{3}} \\ -\frac{1}{\sqrt{6}} \end{pmatrix}$$